

## **The Battle of the Titans (Part II): PROC REPORT versus PROC TABULATE**

Kirk Paul Lafler, sasNerd, Spring Valley, CA

Joshua M. Horstman, Nested Loop Consulting, Indianapolis, IN

Ben Cochran, The Bedford Group, Raleigh, NC

Ray Pass, Gainfully Retired and Having the Time of His Life, St Louis, MO

Dan Bruns, Very Happily Retired

### **Abstract**

Should I use PROC REPORT or PROC TABULATE to produce that report? Which one will give me the control and flexibility to produce the report exactly the way I want it to look? Which one is easier to use? Which one is more powerful? WHICH ONE IS BETTER?

If you have these and other questions about the pros and cons of the REPORT and TABULATE procedures, this presentation is for you. We will discuss, using real-life report scenarios, the strengths (and even a few weaknesses) of the two most powerful reporting procedures in SAS® (as we see it). We will provide you with the knowledge you need to make that difficult decision about which procedure to use to get the report you really want and need.

### **Introduction**

Well, here we go again “Titans” fans – due to the incredible persuasion of the “ORIGINAL” Titan champions, Ray Pass and Dan Bruns, three new warriors, Kirk, Joshua, and Ben, have united to do battle again using the colossus procedures: PROC REPORT and PROC TABULATE. NOT to replicate the “battle” as was waged at previous SAS® Users Group International (SUGIs) conferences, but rather to turn our knowledge of these two incredibly powerful procedures into a “fun”, and engaging presentation.

We will basically be using some basic code to help you discover the power of these two “titan” procedures. We will walk through several scenarios using one procedure to produce a report then produce the same report with the other procedure. Then we will take the same code and expand upon it in the next scenario. Hopefully you will be enlightened as to where each procedure’s strengths lie. So you can reproduce and experiment with the examples, we chose to use the CARS dataset from the SASHELP library. We chose this dataset because we felt it possessed many of the same attributes of some typical datasets/database tables most reports are generated from (categorical values and actual/summarized values).

### **Data Set Used in Examples**

The examples used throughout this battle of the titans utilizes the dataset, SASHELP.CARS. The SASHELP.CARS dataset consists of 428 observations and 15 variables, illustrated below.

Make	Model	Type	Origin	DriveTrain	MSRP	Invoice	Engine Size	Cylinders	Horsepower	MPG_City	MPG_Highway	Weight	Wheelbase	Length
Acura	MDX	SUV	Asia	All	\$38,945	\$33,337	3.5	6	265	17	23	4451	106	189
Acura	RSX Type S 2dr	Sedan	Asia	Front	\$23,820	\$21,781	2.0	4	200	24	31	2778	101	172
Acura	TSX 4dr	Sedan	Asia	Front	\$26,990	\$24,647	2.4	4	200	22	29	3230	105	183
Acura	TL 4dr	Sedan	Asia	Front	\$33,195	\$30,299	3.2	6	270	20	28	3575	108	188
Acura	3.5 RL 4dr	Sedan	Asia	Front	\$43,755	\$39,014	3.5	6	225	18	24	3880	115	197
Acura	3.5 RL w/Navigation 4dr	Sedan	Asia	Front	\$46,100	\$41,100	3.5	6	225	18	24	3893	115	197
Acura	NSX coupe 2dr manual S	Sports	Asia	Rear	\$89,765	\$79,978	3.2	6	290	17	24	3163	100	174
Audi	A4 1.8T 4dr	Sedan	Europe	Front	\$25,940	\$23,508	1.8	4	170	22	31	3252	104	179
Audi	A41.8T convertible 2dr	Sedan	Europe	Front	\$35,940	\$32,506	1.8	4	170	23	30	3638	105	180
Audi	A4 3.0 4dr	Sedan	Europe	Front	\$31,840	\$28,846	3.0	6	220	20	28	3462	104	179
Audi	A4 3.0 Quattro 4dr manual	Sedan	Europe	All	\$33,430	\$30,366	3.0	6	220	17	26	3583	104	179
Audi	A4 3.0 Quattro 4dr auto	Sedan	Europe	All	\$34,480	\$31,388	3.0	6	220	18	25	3627	104	179
Audi	A6 3.0 4dr	Sedan	Europe	Front	\$36,640	\$33,129	3.0	6	220	20	27	3561	109	192
Audi	A6 3.0 Quattro 4dr	Sedan	Europe	All	\$39,640	\$35,992	3.0	6	220	18	25	3880	109	192
Audi	A4 3.0 convertible 2dr	Sedan	Europe	Front	\$42,490	\$38,325	3.0	6	220	20	27	3814	105	180
Audi	A4 3.0 Quattro convertible 2dr	Sedan	Europe	All	\$44,240	\$40,075	3.0	6	220	18	25	4013	105	180
Audi	A6 2.7 Turbo Quattro 4dr	Sedan	Europe	All	\$42,840	\$38,840	2.7	6	250	18	25	3836	109	192
Audi	A6 4.2 Quattro 4dr	Sedan	Europe	All	\$49,690	\$44,936	4.2	8	300	17	24	4024	109	193
Audi	A8 L Quattro 4dr	Sedan	Europe	All	\$69,190	\$64,740	4.2	8	330	17	24	4399	121	204
Audi	S4 Quattro 4dr	Sedan	Europe	All	\$48,040	\$43,556	4.2	8	340	14	20	3825	104	179
Audi	RS 6 4dr	Sports	Europe	Front	\$84,600	\$78,417	4.2	8	450	15	22	4024	109	191
Audi	TT 1.8 convertible 2dr (coupe)	Sports	Europe	Front	\$35,940	\$32,512	1.8	4	180	20	28	3131	95	159
Audi	TT 1.8 Quattro 2dr (convertible)	Sports	Europe	All	\$37,390	\$33,891	1.8	4	225	20	28	2921	96	159
Audi	TT 3.2 coupe 2dr (convertible)	Sports	Europe	All	\$40,590	\$36,739	3.2	6	250	21	29	3351	96	159
Audi	A6 3.0 Avant Quattro	Wagon	Europe	All	\$40,840	\$37,080	3.0	6	220	18	25	4035	109	192
Audi	S4 Avant Quattro	Wagon	Europe	All	\$49,090	\$44,446	4.2	8	340	15	21	3936	104	179
BMW	X3 3.0i	SUV	Europe	All	\$37,000	\$33,873	3.0	6	225	16	23	4023	110	180
BMW	X5 4.4i	SUV	Europe	All	\$52,195	\$47,720	4.4	8	325	16	22	4824	111	184
BMW	325i 4dr	Sedan	Europe	Rear	\$28,495	\$26,155	2.5	6	184	20	29	3219	107	176
BMW	325Ci 2dr	Sedan	Europe	Rear	\$30,795	\$28,245	2.5	6	184	20	29	3197	107	177
BMW	325Ci convertible 2dr	Sedan	Europe	Rear	\$37,995	\$34,800	2.5	6	184	19	27	3560	107	177

Our focus for this paper and presentation is to explore an assortment of scenarios describing general concepts that users will most likely encounter while using PROC TABULATE and PROC REPORT. For users who may not know the underlying design premise for these two procedures, we have included the following excerpt from earlier papers that were written by Dan Bruns, Ray Pass and Alan Eaton of SAS Institute, the original developer of both the REPORT and TABULATE procedures.

### The REPORT Procedure

**Hierarchies Are Not So Good.** The syntax for TABULATE is a challenge for a lot of users. Some form of simplification was in order. The approach taken in PROC REPORT was to limit the number of dimensions and to support a simpler COLUMN statement. Even that COLUMN statement had to be complex enough to support nestings of ACROSS variables and other elements.

**Give Me the Facts.** It was also necessary to offer detail reports with more formatting options than PROC PRINT afforded. PROC REPORT supports both GROUP and ORDER variables to generate both summary and detail reports.

**Hierarchies Are Not So Good.** Users frequently requested that TABULATE support a “post compute” feature to allow new rows or columns to be calculated from the rows and columns of summary statistics already computed from the data. How to specify symbols for use in expressions was a major stumbling block in TABULATE. The expression terms could be very complicated specifications of a sub-dimension of the TABLE statement. So the generality of TABULATE worked against this opportunity for enhancement. Supporting computed columns was a requirement of REPORT from the beginning. The simpler COLUMN statement made it easier to identify symbols to use in column expressions. With symbols available for parsing it became possible to build an interface between PROC REPORT and the DATA step so that PROC REPORT users can compute new columns using DATA step language and functions.

## The TABULATE Procedure

**Hierarchies Are Good.** The primary goals for PROC TABULATE were to form and present hierarchical tables of summary statistics using the data values of classification variables to define hierarchies. An important model for TABULATE was TPL, a table producing language developed at the Department of Labor. TPL did a wonderful job of defining, building, and presenting complex hierarchical tables. However, TPL required users to predefine a significant amount of structural information in codebooks and could require as many as 15 JCL steps to execute. Our goal was to make the table building process entirely data driven and to simplify the task to writing a few lines of SAS code rather than many lines of complex code.

**Give Me the Summary Version.** PROC SUMMARY already computed summary statistics. TABULATE's job was to organize these summary statistics for presentation. Other SAS procedures were available in the early days for detailed listings, PROC PRINT, and row and column computations, PROC COMPUTAB. TABULATE made no attempt to offer these kinds of features.

**Generality in All Dimensions.** PROC TABULATE supports three-dimensional tables, tables with pages, rows, and columns. TABULATE was designed to support arbitrarily complex nestings and concatenations of table elements in each dimension. So TABULATE provides a lot of flexibility but at the cost of seeming overly complex and difficult to master at times.

## Round #1

Many users produce detail output and reports on their data, we will begin here. (We will examine how a summary report can be produced in Round #2). In this round, we will examine the REPORT and TABULATE procedures to produce detail output / reports. Several statements and options are available to users of PROC REPORT and PROC TABULATE to produce detail output / reports. For the REPORT procedure, the COLUMN(S) and DEFINE statements can be specified; and for the TABULATE procedure, the CLASS, VAR, and TABLE statements can be specified; see below.

### PROC REPORT Code:

```
TITLE1 Detailed Cars Report <= $13,000 ;
TITLE2 Arranged by Origin ;
TITLE3 Produced by PROC REPORT ;
PROC REPORT DATA=sashelp.cars ;
  COLUMNS origin type make model MSRP ;
  DEFINE origin / ORDER 'Origin of Car' WIDTH=6 ;
  DEFINE type / ORDER 'Type of Car' WIDTH=8 CENTER ;
  DEFINE make / ORDER 'Make of Car' WIDTH=13 ;
  DEFINE model / DISPLAY 'Model of Car' WIDTH=40 ;
  DEFINE MSRP / DISPLAY WIDTH=8 FORMAT=Dollar8. ;
  WHERE MSRP LE 13000 ;
RUN ;
```

**Results:**

Detailed Cars Report <= \$13,000 Arranged by Origin Produced by PROC REPORT						
Origin of Car	Type of Car	Make of Car	Model of Car	MSRP		
Asia	Sedan	Hyundai	Accent 2dr hatch	\$10,539		
			Accent GL 4dr	\$11,839		
			Accent GT 2dr hatch	\$11,939		
			Kia	Rio 4dr manual	\$10,280	
				Rio 4dr auto	\$11,155	
				Spectra 4dr	\$12,360	
			Nissan	Sentra 1.8 4dr	\$12,740	
				Scion	xA 4dr hatch	\$12,965
				Suzuki	Aeno S 4dr	\$12,884
			Toyota	Forenza S 4dr	\$12,269	
				Echo 2dr manual	\$10,760	
				Echo 2dr auto	\$11,560	
			Echo 4dr	\$11,290		
			Truck	Toyota	Tacoma	\$12,800
USA	Wagon	Kia	Rio Cinco	\$11,905		
			Chevrolet	Aveo 4dr	\$11,690	
						Aveo LS 4dr hatch
		Saturn	Ion1 4dr	\$10,995		

**PROC TABULATE Code:**

```

TITLE1 Detailed Cars Report <= $13,000 ;
TITLE2 Arranged by Origin ;
TITLE3 Produced by PROC TABULATE ;
PROC TABULATE DATA=sashelp.cars FORMAT=DOLLAR8. ;
CLASS origin type make model ;
TABLE origin * type *
      make * model,
      MSRP ;
VAR MSRP ;
WHERE MSRP LE 13000 ;
RUN ;
    
```

**Results:**

**Detailed Cars Report <= \$13,000  
Arranged by Origin  
Produced by PROC TABULATE**

				MSRP
				Sum
Origin	Type	Make	Model	
Asia	Sedan	Hyundai	Accent 2dr hatch	\$10,539
			Accent GL 4dr	\$11,839
			Accent GT 2dr hatch	\$11,939
		Kia	Rio 4dr auto	\$11,155
			Rio 4dr manual	\$10,280
			Spectra 4dr	\$12,360
		Nissan	Sentra 1.8 4dr	\$12,740
		Scion	xA 4dr hatch	\$12,965
		Suzuki	Aeno S 4dr	\$12,884
			Forenza S 4dr	\$12,269
		Toyota	Echo 2dr auto	\$11,560
			Echo 2dr manual	\$10,760
			Echo 4dr	\$11,290
		Truck	Toyota	Tacoma
Wagon	Kia	Rio Cinco	\$11,905	
USA	Sedan	Chevrolet	Aveo 4dr	\$11,690
			Aveo LS 4dr hatch	\$12,585
		Saturn	Ion1 4dr	\$10,995

**Round #2**

In the second round we will turn our attention from the production of detail output / reports to summary output / reports. In other words, we will describe and illustrate the specific statements and options to produce summary-level output / reports that display several descriptive statistics (i.e., SUM, N, MEAN, MIN, and MAX); using PROC REPORT and PROC TABULATE.

**PROC REPORT Code:**

```

TITLE1 MSRP Summary Statistics Report ;
TITLE2 Cars <= $18,000 ;
TITLE3 Produced by PROC REPORT ;
PROC REPORT DATA=sashep.cars ;
    COLUMNS type make MSRP MSRP=MSRP_Sum MSRP=MSRP_N
              MSRP=MSRP_Mean MSRP=MSRP_Min MSRP=MSRP_Max ;
    DEFINE type / GROUP 'Type of Car' WIDTH=8 ;
    DEFINE make / GROUP 'Make of Car' WIDTH=13 ;
    DEFINE MSRP / ANALYSIS NOPRINT FORMAT=Dollar12. ;
    DEFINE MSRP_SUM / ANALYSIS SUM FORMAT=Dollar12. "Sum" ;
    DEFINE MSRP_N / ANALYSIS N "N" FORMAT=COMMA7. ;
    DEFINE MSRP_MEAN / ANALYSIS MEAN "Mean" ;
    
```

```

DEFINE MSRP_MIN / ANALYSIS MIN "Min" ;
DEFINE MSRP_MAX / ANALYSIS MAX "Max" ;
WHERE MSRP LE 18000 ;
RUN ;
    
```

**Results:**

**MSRP Summary Statistics Report**  
Cars <= \$18,000  
Produced by PROC REPORT

Type of Car	Make of Car	Sum	N	Mean	Min	Max
SUV	Suzuki	\$17,163	1	\$17,163	\$17,163	\$17,163
Sedan	Chevrolet	\$70,080	5	\$14,016	\$11,690	\$18,385
	Chrysler	\$17,985	1	\$17,985	\$17,985	\$17,985
	Dodge	\$28,710	2	\$14,355	\$13,670	\$15,040
	Ford	\$58,040	4	\$14,510	\$13,270	\$15,580
	Honda	\$61,040	4	\$15,260	\$13,270	\$17,750
	Hyundai	\$78,934	6	\$13,156	\$10,539	\$15,389
	Kia	\$78,045	6	\$13,008	\$10,280	\$16,040
	MINI	\$16,999	1	\$16,999	\$16,999	\$16,999
	Mazda	\$32,700	2	\$16,350	\$15,500	\$17,200
	Mitsubishi	\$48,576	3	\$16,192	\$14,622	\$17,232
	Nissan	\$45,120	3	\$15,040	\$12,740	\$17,640
	Pontiac	\$33,230	2	\$16,615	\$15,495	\$17,735
	Saturn	\$72,320	5	\$14,464	\$10,995	\$18,350
	Scion	\$12,965	1	\$12,965	\$12,965	\$12,965
	Suzuki	\$72,483	5	\$14,497	\$12,269	\$17,262
	Toyota	\$78,020	6	\$13,003	\$10,780	\$15,295
Truck	Dodge	\$17,630	1	\$17,630	\$17,630	\$17,630
	Ford	\$14,385	1	\$14,385	\$14,385	\$14,385
	GMC	\$16,530	1	\$16,530	\$16,530	\$16,530
	Mazda	\$14,840	1	\$14,840	\$14,840	\$14,840
	Toyota	\$29,295	2	\$14,648	\$12,800	\$16,495
Wagon	Ford	\$17,475	1	\$17,475	\$17,475	\$17,475
	Kia	\$11,905	1	\$11,905	\$11,905	\$11,905
	Mitsubishi	\$17,495	1	\$17,495	\$17,495	\$17,495
	Pontiac	\$17,045	1	\$17,045	\$17,045	\$17,045
	Scion	\$14,165	1	\$14,165	\$14,165	\$14,165
	Suzuki	\$16,497	1	\$16,497	\$16,497	\$16,497
	Toyota	\$16,695	1	\$16,695	\$16,695	\$16,695

**PROC TABULATE Code:**

```

TITLE1 MSRP Summary Statistics Report ;
TITLE2 Cars <= $18,000 ;
TITLE3 Produced by PROC TABULATE ;
PROC TABULATE DATA=sashe1p.cars ;
    CLASS type make ;
    TABLE type * make,
           MSRP * (SUM N MEAN MIN MAX) ;
VAR MSRP ;
WHERE MSRP LE 18000 ;
RUN ;
    
```

**Results:**

**MSRP Summary Statistics Report**  
Cars <= \$18,000  
Produced by PROC TABULATE

		MSRP				
		Sum	N	Mean	Min	Max
Type	Make					
SUV	Suzuki	17163.00	1	17163.00	17163.00	17163.00
Sedan	Chevrolet	70080.00	5	14016.00	11690.00	16385.00
	Chrysler	17985.00	1	17985.00	17985.00	17985.00
	Dodge	28710.00	2	14355.00	13670.00	15040.00
	Ford	59040.00	4	14510.00	13270.00	15580.00
	Honda	61040.00	4	15260.00	13270.00	17750.00
	Hyundai	78934.00	6	13155.67	10539.00	15389.00
	Kia	78045.00	6	13007.50	10280.00	16040.00
	MINI	16999.00	1	16999.00	16999.00	16999.00
	Mazda	32700.00	2	16350.00	15500.00	17200.00
	Mitsubishi	48578.00	3	16192.00	14622.00	17232.00
	Nissan	45120.00	3	15040.00	12740.00	17640.00
	Pontiac	33230.00	2	16615.00	15495.00	17735.00
	Saturn	72320.00	5	14464.00	10995.00	16350.00
	Scion	12965.00	1	12965.00	12965.00	12965.00
	Suzuki	72483.00	5	14496.60	12269.00	17262.00
Toyota	78020.00	6	13003.33	10780.00	15295.00	
Truck	Dodge	17630.00	1	17630.00	17630.00	17630.00
	Ford	14385.00	1	14385.00	14385.00	14385.00
	GMC	16530.00	1	16530.00	16530.00	16530.00
	Mazda	14840.00	1	14840.00	14840.00	14840.00
	Toyota	29295.00	2	14647.50	12800.00	16495.00
Wagon	Ford	17475.00	1	17475.00	17475.00	17475.00
	Kia	11905.00	1	11905.00	11905.00	11905.00
	Mitsubishi	17495.00	1	17495.00	17495.00	17495.00
	Pontiac	17045.00	1	17045.00	17045.00	17045.00
	Scion	14185.00	1	14185.00	14185.00	14185.00
	Suzuki	16497.00	1	16497.00	16497.00	16497.00
	Toyota	16695.00	1	16695.00	16695.00	16695.00

**Round #3**

In the third round we will use PROC REPORT and PROC TABULATE, along with the Output Delivery System (ODS) Excel destination, to send output / reports to a Microsoft® Excel spreadsheet. We describe and illustrate the SAS-supplied style templates, the ODS Excel destination, and options, below.

**PROC TEMPLATE List Styles Code:**

```
PROC TEMPLATE;
  LIST STYLES;
RUN;
```

The results produced from the PROC TEMPLATE LIST STYLES statement produces the following styles, see below.

**Results:**

Listing of: SASHELP.TMPLMST		
Path Filter is: Styles		
Sort by: PATH/ASCENDING		
Obs	Path	Type
1	Styles	Dir
2	Styles.Analysis	Style
3	Styles.BarrettsBlue	Style
4	Styles.BlockPrint	Style
5	Styles.DTree	Style
6	Styles.Daisy	Style
7	Styles.Default	Style
8	Styles.Dove	Style
9	Styles.EGDefault	Style
10	Styles.Excel	Style
11	Styles.FancyPrinter	Style
12	Styles.Festival	Style
13	Styles.FestivalPrinter	Style
14	Styles.Gantt	Style
15	Styles.GrayscalePrinter	Style
16	Styles.HTMLBlue	Style
17	Styles.Harvest	Style
18	Styles.HighContrast	Style

19	Styles.HighContrastLarge	Style
20	Styles.Journal	Style
21	Styles.Journal1a	Style
22	Styles.Journal2	Style
23	Styles.Journal2a	Style
24	Styles.Journal3	Style
25	Styles.Journal3a	Style
26	Styles.Listing	Style
27	Styles.Meadow	Style
28	Styles.MeadowPrinter	Style
29	Styles.Minimal	Style
30	Styles.MonochromePrinter	Style
31	Styles.Monospace	Style
32	Styles.Moonflower	Style
33	Styles.Netdraw	Style
34	Styles.NoFontDefault	Style
35	Styles.Normal	Style
36	Styles.NormalPrinter	Style
37	Styles.Ocean	Style
38	Styles.Pearl	Style
39	Styles.PearlJ	Style
40	Styles.Plateau	Style

41	Styles.PowerPointDark	Style
42	Styles.PowerPointLight	Style
43	Styles.Printer	Style
44	Styles.Raven	Style
45	Styles.Rtf	Style
46	Styles.Sapphire	Style
47	Styles.SasDocPrinter	Style
48	Styles.SasWeb	Style
49	Styles.Seaside	Style
50	Styles.SeasidePrinter	Style
51	Styles.Snow	Style
52	Styles.StatDoc	Style
53	Styles.Statistical	Style
54	Styles.Word	Style
55	Styles.vaDark	Style
56	Styles.vaHighContrast	Style
57	Styles.vaLight	Style

The next example illustrates the creation of an Excel spreadsheet by specifying the ODS Excel destination with PROC REPORT.

**PROC REPORT Code:**

```

ODS Excel FILE='REPORT-to-Excel.XLSX'
      STYLE=BarrettsBlue ;

TITLE1 Detailed Cars Report <= $13,000 ;
TITLE2 Arranged by Origin ;
TITLE3 Produced by PROC REPORT ;
PROC REPORT DATA=sashelp.cars ;
  COLUMNS origin type make model MSRP ;
  DEFINE origin / ORDER 'Origin of Car' WIDTH=6 ;
  DEFINE type / ORDER 'Type of Car' WIDTH=8 CENTER ;
  DEFINE make / ORDER 'Make of Car' WIDTH=13 ;
  DEFINE model / DISPLAY 'Model of Car' WIDTH=40 ;
  DEFINE MSRP / DISPLAY WIDTH=8 FORMAT=Dollar8. ;
  WHERE MSRP LE 13000 ;

RUN ;
ODS Excel CLOSE ;
    
```



**Results:**

	A	B	C	D	E
1	<b>Origin of Car</b>	<b>Type of Car</b>	<b>Make of Car</b>	<b>Model of Car</b>	<b>MSRP</b>
2	Asia	Sedan	Hyundai	Accent 2dr hatch	\$10,539
3				Accent GL 4dr	\$11,839
4				Accent GT 2dr hatch	\$11,939
5			Kia	Rio 4dr manual	\$10,280
6				Rio 4dr auto	\$11,155
7				Spectra 4dr	\$12,360
8			Nissan	Sentra 1.8 4dr	\$12,740
9			Scion	xA 4dr hatch	\$12,965
10			Suzuki	Aeno S 4dr	\$12,884
11				Forenza S 4dr	\$12,269
12			Toyota	Echo 2dr manual	\$10,760
13				Echo 2dr auto	\$11,560
14				Echo 4dr	\$11,290
15		Truck	Toyota	Tacoma	\$12,800
16		Wagon	Kia	Rio Cinco	\$11,905
17	USA	Sedan	Chevrolet	Aveo 4dr	\$11,690
18				Aveo LS 4dr hatch	\$12,585
19			Saturn	Ion1 4dr	\$10,995

The next example illustrates the creation of an Excel spreadsheet by specifying the ODS Excel destination with PROC TABULATE.

**PROC TABULATE Code:**

```

ODS Excel FILE='TABULATE-to-Excel.XLSX'
      STYLE=BarrettsBlue ;
TITLE1 Detailed Cars Report <= $13,000 ;
TITLE2 Arranged by Origin ;
TITLE3 Produced by PROC TABULATE ;
PROC TABULATE DATA=saselp.cars FORMAT=DOLLAR8. ;
  CLASS origin type make model ;
  TABLE origin * type *
         make * model,
         MSRP ;
  VAR MSRP ;
  WHERE MSRP LE 13000 ;
RUN ;
ODS Excel CLOSE ;

```

**Results:**

	A	B	C	D	E
1					<b>MSRP</b>
2					<b>Sum</b>
3	<b>Origin</b>	<b>Type</b>	<b>Make</b>	<b>Model</b>	
4	Asia	Sedan	Hyundai	Accent 2dr hatch	\$10,539
5				Accent GL 4dr	\$11,839
6				Accent GT 2dr hatch	\$11,939
7			Kia	Rio 4dr auto	\$11,155
8				Rio 4dr manual	\$10,280
9				Spectra 4dr	\$12,360
10			Nissan	Sentra 1.8 4dr	\$12,740
11			Scion	xA 4dr hatch	\$12,965
12			Suzuki	Aeno S 4dr	\$12,884
13				Forenza S 4dr	\$12,269
14			Toyota	Echo 2dr auto	\$11,560
15				Echo 2dr manual	\$10,760
16				Echo 4dr	\$11,290
17		Truck	Toyota	Tacoma	\$12,800
18		Wagon	Kia	Rio Cinco	\$11,905
19	USA	Sedan	Chevrolet	Aveo 4dr	\$11,690
20				Aveo LS 4dr hatch	\$12,585
21			Saturn	Ion1 4dr	\$10,995

**Round #4**

In the fourth, and final, round we will illustrate various style options with PROC REPORT and PROC TABULATE code to enhance the look of the output / report. What and how would you change to the code to make it more presentable? What statements, options and features could be specified to improve the appearance of the output / report? In the following table several style components are illustrated using PROC REPORT and PROC TABULATE, see below.

<b>Component</b>	<b>Description</b>
<b>Report</b>	Affects the report and the table structure.
<b>Header (HDR)</b>	Affects color header cells.
<b>Column</b>	Affects data cells.
<b>Summary</b>	Affects summary lines generated by BREAK or RBREAK statements.
<b>Lines</b>	Affects lines generated by LINE statements.

Specific style attributes available to PROC REPORT and PROC TABULATE are illustrated, see below.

Style Attribute	Description
<b>Background=</b>	Changes background color (e.g., Red, Blue, Green, Orange, Yellow).
<b>Bordercolor=</b>	Changes border color (e.g., Red, Blue, Green, Orange, Yellow).
<b>Borderwidth=</b>	Changes border width (e.g., 0, 3, 7, in, cm, pt).
<b>Cellspacing=</b>	Changes cell spacing (e.g., 0, 3, 7, in, cm, pt).
<b>Foreground=</b>	Changes foreground color (e.g., Red, Blue, Green, Orange, Yellow).
<b>Frame=</b>	Changes frame style (e.g., ABOVE, BELOW, BOX, HSIDES, LHS, RHS, VOID and VSIDES).
<b>Rules=</b>	Changes rules (e.g., ALL, COLS, GROUPS, NONE, and ROWS).

In the next PROC REPORT code we examine how labels, titles, formats, and the style= option can be specified to enhance the output / report aesthetics.

**PROC REPORT Code:**

```

TITLE1 Customized Cars Report <= $13,000 ;
TITLE2 with Style ;
TITLE3 Produced by PROC REPORT ;
PROC REPORT DATA=sashelp.cars ;
  COLUMNS origin type make model MSRP ;
  DEFINE origin / ORDER 'Origin of Car' WIDTH=6 ;
  DEFINE type / ORDER 'Type of Car' WIDTH=8 CENTER ;
  DEFINE make / ORDER 'Make of Car' WIDTH=13 ;
  DEFINE model / DISPLAY 'Model of Car' WIDTH=40 ;
  DEFINE MSRP / DISPLAY WIDTH=8 FORMAT=Dollar8.
  STYLE(CoLumn)={Background=Red
                  Foreground=White} ;
WHERE MSRP LE 13000 ;
RUN ;

```

**Results:**

Customized Cars Report <= \$13,000 with Style Produced by PROC REPORT							
Origin of Car	Type of Car	Make of Car	Model of Car	MSRP			
Asia	Sedan	Hyundai	Accent 2dr hatch	\$10,539			
			Accent GL 4dr	\$11,839			
			Accent GT 2dr hatch	\$11,939			
			Kia	Rio 4dr manual	\$10,280		
				Rio 4dr auto	\$11,155		
				Spectra 4dr	\$12,360		
			Nissan	Sentra 1.8 4dr	\$12,740		
				Scion	xA 4dr hatch	\$12,965	
				Suzuki	Aeno S 4dr	\$12,884	
			Suzuki	Forenza S 4dr	\$12,269		
				Toyota	Echo 2dr manual	\$10,760	
						Echo 2dr auto	\$11,560
						Echo 4dr	\$11,290
	Truck	Toyota	Tacoma	\$12,800			
	Wagon	Kia	Rio Cinco	\$11,905			
USA	Sedan	Chevrolet	Aveo 4dr	\$11,690			
			Aveo LS 4dr hatch	\$12,585			
		Saturn	Ion1 4dr	\$10,995			

In the next PROC TABULATE code we examine how labels, titles, formats, and the style= option can be specified to enhance the output / report aesthetics.

**PROC TABULATE Code:**

```

TITLE1 Customized Cars Report <= $13,000 ;
TITLE2 with Style ;
TITLE3 Produced by PROC TABULATE ;
PROC TABULATE DATA=sashep.cars FORMAT=DOLLAR8. ;
CLASS origin type make model ;
TABLE origin * type *
      make * model,
      MSRP * {style={Background=red
                  Foreground=White}} ;
VAR MSRP ;
WHERE MSRP LE 13000 ;
RUN ;

```

**Results:**

**Customized Cars Report <= \$13,000  
with Style  
Produced by PROC TABULATE**

				MSRP
				Sum
Origin	Type	Make	Model	
Asia	Sedan	Hyundai	Accent 2dr hatch	\$10,539
			Accent GL 4dr	\$11,839
			Accent GT 2dr hatch	\$11,939
		Kia	Rio 4dr auto	\$11,155
			Rio 4dr manual	\$10,280
			Spectra 4dr	\$12,360
		Nissan	Sentra 1.8 4dr	\$12,740
		Scion	xA 4dr hatch	\$12,965
		Suzuki	Aeno S 4dr	\$12,884
			Forenza S 4dr	\$12,269
	Toyota	Echo 2dr auto	\$11,560	
		Echo 2dr manual	\$10,760	
		Echo 4dr	\$11,290	
Truck	Toyota	Tacoma	\$12,800	
Wagon	Kia	Rio Cinco	\$11,905	
USA	Sedan	Chevrolet	Aveo 4dr	\$11,690
			Aveo LS 4dr hatch	\$12,585
		Saturn	Ion1 4dr	\$10,995

In the next PROC REPORT code we examine how labels, titles, formats, and the style= option can be specified to enhance the output / report aesthetics by displaying traffic lighting scenarios as a “background” color.

**PROC REPORT Code:**

```
TITLE1 Customized Cars Report ;
TITLE2 PROC REPORT with Traffic Lighting ;
PROC FORMAT ;
    VALUE MSRPFmt LOW - < 20000 = 'Green'
           20000 - < 30000 = 'Blue'
           30000 - < 40000 = 'Orange'
           40000 - HIGH    = 'Red' ;
run ;
```

```
PROC REPORT DATA=SASHELP.Cars
            STYLE(Header)={BackGround=Blue ForeGround=White
                           Font=(Arial, 10pt, Bold)} ;
```

```
COLUMNS Origin Make Type Model MSRP ;
DEFINE Origin / ORDER 'Country of Origin' WIDTH=6 ;
DEFINE Make / ORDER 'Make of Vehicle' WIDTH=13 ;
DEFINE Type / ORDER 'Vehicle Type' WIDTH=8 ;
DEFINE Model / DISPLAY 'Vehicle Model' WIDTH=40 ;
```

```

DEFINE MSRP / ANALYSIS 'Vehicle MSRP' WIDTH=8
STYLE(Column)=[FontWeight=Bold
                ForeGround=White
                BackGround=MSRPFmt.] ;
    
```

RUN ;

**Results:**

**Customized Cars Report  
PROC REPORT with Traffic Lighting**

Country of Origin	Make of Vehicle	Vehicle Type	Vehicle Model	Vehicle MSRP	
Asia	Acura	SUV	MDX	\$36,945	
		Sedan	RSX Type S 2dr	\$23,820	
			TSX 4dr	\$26,990	
			TL 4dr	\$33,195	
			3.5 RL 4dr	\$43,755	
			3.5 RL w/Navigation 4dr	\$46,100	
			Sports	NSX coupe 2dr manual S	\$89,765
		Honda	Hybrid	Civic Hybrid 4dr manual (gas/electric)	\$20,140
				Insight 2dr (gas/electric)	\$19,110
			SUV	Pilot LX	\$27,560
CR-V LX	\$19,860				
Element LX	\$18,690				
Sedan	Civic DX 2dr			\$13,270	
	Civic HX 2dr			\$14,170	
	Civic LX 4dr			\$15,850	
	Accord LX 2dr			\$19,860	
	Accord EX 2dr			\$22,260	
	Civic EX 4dr	\$17,750			
	Civic Si 2dr hatch	\$19,490			
	Accord LX V6 4dr	\$23,760			
Accord EX V6 2dr	\$26,960				
			Odyssey LX	\$24,950	
			Odyssey EX	\$27,450	
			Sports	S2000 convertible 2dr	\$33,260
	Hyundai	SUV	Santa Fe GLS	\$21,589	

In the next PROC TABULATE code we examine how labels, titles, formats, and the style= option can be specified to enhance the output / report aesthetics by displaying traffic lighting scenarios.

**PROC TABULATE Code:**

```

TITLE1 Customized Cars Report ;
TITLE2 PROC TABULATE with Traffic Lighting ;
PROC FORMAT ;
    VALUE MSRPFmt LOW - < 20000 = 'Green'
                20000 - < 30000 = 'Blue'
                30000 - < 40000 = 'Orange'
                40000 - HIGH = 'Red' ;
run ;
    
```

```

PROC TABULATE DATA=SASHELP.Cars
  FORMAT=DOLLAR10.
  STYLE={BackGround=Blue ForeGround=White
        Font=(Arial, 10pt, Bold)} ;
CLASS origin type make model / Style={BackGround=Blue
        ForeGround=White} ;
TABLE origin * type * make * model,
  MSRP * {STYLE=[FontWeight=Bold
        ForeGround=White
        BackGround=MSRPFmt.]} ;

```

VAR MSRP ;

RUN ;

**Results:**

**Customized Cars Report  
PROC TABULATE with Traffic Lighting**

				MSRP		
				Sum		
Origin	Type	Make	Model			
Asia	Hybrid	Honda	Civic Hybrid 4dr manual (gas/electric)	\$20,140		
			Insight 2dr (gas/electric)	\$19,110		
		Toyota	Prius 4dr (gas/electric)	\$20,510		
	SUV	Acura	MDX		\$36,945	
				Honda	CR-V LX	\$19,860
					Element LX	\$18,690
		Pilot LX	\$27,560			
		Hyundai	Santa Fe GLS	\$21,589		
		Isuzu	Ascender S		\$31,849	
				Rodeo S	\$20,449	
		Kia	Sorento LX	\$19,635		
		Lexus	GX 470		\$45,700	
				LX 470	\$64,800	
				RX 330	\$39,195	
		Mazda	Tribute DX 2.0	\$21,087		
		Mitsubishi	Endeavor XLS		\$30,492	
				Montero XLS	\$33,112	
				Outlander LS	\$18,892	
		Nissan	Pathfinder Armada SE		\$33,840	
				Pathfinder SE	\$27,339	
Xterra XE V6	\$20,939					
Suzuki	Vitara LX		\$17,163			
		XL-7 EX	\$23,699			

## Conclusion

So, “Which colossus procedure is better?”, you might ask. Well, our task was not to declare a winner or a loser, but rather to show that PROC REPORT and PROC TABULATE could be used to solve your reporting needs. This paper / presentation is intended to illustrate the subtle differences between PROC REPORT and PROC TABULATE. What we have attempted to do is give you a head start, and hopefully ignite an excitement for further discovery about these two SAS “TITANS” procedures. (To be clear, we’re talking about the procedures here, not the authors.) It is also worth mentioning that each procedure has a definite set of strengths over the other depending on the desired result. The tools are there and they serve as powerful foundations to valuable and productive report generation.

## References

- Carpenter, Arthur L. (2016). [\*“Color, Rank, Count, Name; Controlling it all in PROC REPORT,”\*](#) Proceedings of the 2016 SAS Global Forum (SGF) Conference.
- Cochran, Ben (2005). [\*“A Gentle Introduction to the Powerful REPORT Procedure,”\*](#) Proceedings of the 2005 SAS Users Group International (SUGI) Conference.
- Lafler, Kirk Paul, Joshua Horstman, Ben Cochran, Ray Pass, and Dan Bruns (2022). [\*“Battle of the Titans \(Part II\): PROC REPORT versus PROC TABULATE,”\*](#) Proceedings of the 2022 SouthEast SAS Users Group (SESUG) Conference.
- Lafler, Kirk Paul, Joshua Horstman, Ben Cochran, Ray Pass, and Dan Bruns (2022). [\*“Battle of the Titans \(Part II\): PROC REPORT versus PROC TABULATE,”\*](#) Proceedings of the 2022 Western Users of SAS Software (WUSS) Conference.
- Lafler, Kirk Paul, Ben Cochran, Josh Horstman and Ray Pass (2019). [\*“Battle of the Titans \(Part II\): PROC TABULATE versus PROC REPORT,”\*](#) Proceedings of the 2019 SouthEast SAS Users Group (SESUG) Conference.
- Lafler, Kirk Paul, Ben Cochran and Ray Pass (2017). [\*“Battle of the Titans \(Part II\): PROC TABULATE versus PROC REPORT,”\*](#) Proceedings of the 2017 MidWest SAS Users Group (MWSUG) Conference.
- Lafler, Kirk Paul (2016). [\*“An Introduction to PROC REPORT,”\*](#) Proceedings of the 2016 SouthEast SAS Users Group (SESUG) Conference.
- Lafler, Kirk Paul (2016). [\*“An Introduction to PROC REPORT,”\*](#) Proceedings of the 2016 Nebraska SAS Users Group (NEBSUG) Conference.
- LeBouton, Kimberly (2014). [\*“Getting Up to Speed with PROC REPORT,”\*](#) Proceedings of the 2014 Western Users of SAS Software (WUSS) Conference.
- Lewandowski, David (2008). [\*“A Step-by-Step Introduction to PROC REPORT,”\*](#) Proceedings of the 2008 SAS Global Forum (SGF) Conference.
- Pass, Ray and Sandy McNeill (2006). [\*“PROC REPORT: Doin’ It in Style,”\*](#) Proceedings of the 2006 SAS Users Group International (SUGI) Conference.

## Acknowledgments

The authors thank the PharmaSUG 2023 Conference Committee, particularly the Solution Development Section Chairs, Ajay Gupta and Renuka Tammisetti, for accepting our abstract and paper; the PharmaSUG 2023 Executive Committee for organizing and supporting a “live” conference event; SAS Institute Inc. for providing SAS users with wonderful software; and SAS users everywhere for being the nicest people anywhere!

## Trademark Citations

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies.



## About the Authors

Kirk Paul Lafler is a SAS, SQL, Python, Excel, and data analytics educator, developer, programmer, and consultant; a lecturer and adjunct professor at San Diego State University (SDSU); an adjunct professor at University of California San Diego (UCSD) Extension; and teaches SAS, SQL, Python, Excel, and data analytics courses, workshops, and webinars to users around the world. Kirk has been a SAS user since 1979 and is the author of several books including, PROC SQL: Beyond the Basics Using SAS, Third Edition (SAS Press, 2019) along with papers and articles on a variety of SAS topics. Kirk has participated as an Invited speaker, educator, keynote, and section leader at SAS conferences and education forums; and is the recipient of 27 “Best” contributed paper, hands-on workshop (HOW), and poster awards.

Joshua M. Horstman is an independent statistical programmer with 18 years’ experience using SAS in the pharmaceutical industry. He has experience working for GlaxoSmithKline, Eli Lilly and Company, inVentiv Health, and others. Josh has presented numerous papers at PharmaSUG, SAS Global Forum, and other SAS User Group meetings. He holds a bachelor's degree in mathematics and computer science, and a master's degree in statistics from Colorado State University.

After more than 11 years with SAS Institute in the Professional Services (as an Instructor) and Marketing Departments (as Marketing Manager for the SAS/EIS product), Ben Cochran left to start his own consulting and SAS Training business in the fall of 1996 – The Bedford Group. As an affiliate member of SAS Institute’s Alliance Partner Program, Ben has been involved in many teaching and consulting projects over the last 20+ years. Ben has authored and presented several papers at SUGI, SGF, and regional user groups on a variety of topics since 1988.

After 35 years or so, Ray Pass, one of the original champions of the Battle of the Titans, retired on June 1, 2016 to be exact, (but who cares) from full-time SASing. He now spends his time living, thinking, reading, cooking, eating, walking, biking, shopping, erranding, and as much flaneuring as he can. After recently coming out of retirement he is having the time of his life solving SAS puzzles, teaming up with a great bunch of people with lots of mutual respect and getting paid for it - couldn’t be better. He has adopted a new life motto: **"Every day is Saturday!"**

Dan Bruns – A legendary SAS user for many years and one of the original champions of Battle of the Titans.

Comments and suggestions can be sent to:

Kirk Paul Lafler

SAS® / SQL / Python / Excel / Data Analytics Educator, Developer, Programmer, Consultant, Data Analyst, and Author  
sasNerd

E-mail: [KirkLafler@cs.com](mailto:KirkLafler@cs.com)

LinkedIn: <http://www.linkedin.com/in/KirkPaulLafler>

Twitter: @sasNerd

Joshua M. Horstman

Senior SAS® Consultant, Application Developer and Author

E-mail: [jmhorstman@gmail.com](mailto:jmhorstman@gmail.com)

LinkedIn: <https://www.linkedin.com/in/joshuahorstman>

Ben Cochran

SAS® Consultant, Application Developer, Programmer and Educator  
The Bedford Group

E-mail: [bentcochran@gmail.com](mailto:bentcochran@gmail.com)

LinkedIn: <https://www.linkedin.com/in/ben-cochran-35577112/>

Dr. Ray Pass

SAS® Consultant, Programmer, Data Analyst and Author

“Retired and Having the Time of His Life”

E-mail: [RayPass@att.net](mailto:RayPass@att.net)

LinkedIn: <https://www.linkedin.com/in/ray-pass-phd-152781b/>

Dan Bruns  
SAS® Programmer, Data Analyst and Author  
“Very Happily Retired”  
E-mail: [danielelbruns@gmail.com](mailto:danielelbruns@gmail.com)